
ABSTRACT
A macro variable array is a jargon term for a list of macro variables with a common prefix and
numerical suffixes. Macro arrays are valued by advanced SAS○R programmers and often used as
"driving" lists, allowing sequential metadata for complex or iterative programs. Use of macro
arrays requires advanced macro programming techniques based on indirect reference (aka, using
multiple ampersands &&), which may intimidate less experienced programmers. The aim of the
paper is to introduce the macroArray SAS package. The package facilitates a solution that
makes creation and work with macro arrays much easier. It also provides a "DATA-step-arrays-like"
interface that allows use of macro arrays without complications that arise from indirect referencing.
Also, the concept of a macro dictionary is presented, and all concepts are demonstrated through
use cases and examples.

INTRODUCTION

A macro variable array (also MVA or just a macro array) is a jargon term for a list of macro variables with a com-
mon prefix and numerical sequential suffixes. Usually such terms have a form similar to&&prefix&suffix
(e.g., &&var&i).

Every year for the last 25 years (i.e., 2000 - 2024), there has been at least one paper related to the
subject of macro variable arrays published in at least one major SAS conference proceedings (this paper
included). For example: [Widawski 2000], [First 2001], [Widawski 2002], [Carpenter & Smith 2003],
[Fehd 2003], [Clay 2004], [Carpenter 2005], [Clay 2006], [Lavery 2007], [Philp 2008], [Rosson 2009],
[Russell & Tyndall 2010], [Gilsen 2011], [Li 2012], [Zender 2013], [Werner 2014], [Nayak 2015],
[Wong & Short 2016], [Carpenter 2017], [Renauldo 2018], [Horstman 2019], [Huang 2020],
[Wang 2021], [Walker 2022], and [Roudneva 2023]. With a quick glimpse at www.lexjansen.com, it
can be seen that articles pertaining to macro variable arrays spread they range from: simple examples, by
detailed explanations and tutorials, to use cases, and eventually to development of tools to make work with
MVAs easier and more comfortable.

On the one hand, MVAs (or just macro arrays) are valued by advanced SAS programmers and are often used
as "driving" lists allowing to provide sequential metadata for complex or iterative programs. On the other,

∗The text was originally published in PharmaSUG 2024 proceedings (see [Jablonski 2024]). This is a republish.

use of macro arrays requires advance macro programming techniques based on indirect reference, aka work
with multiple ampersands (&&), which may be intimidating for less experienced programmers.

The aim of the article is to introduce macroArray SAS package. The package facilitates set of SAS macros
which make creation and work with macro arrays much easier. A "data-step-arrays-like" interface, which
allows use of macro arrays without complications which arise from indirect referencing, is provided. Also an
idea of macro dictionary is presented. So, according to the classification mentioned above, this article falls
into the "development of tools to make work with MVAs easier and more comfortable" category.

MACRO VARIABLE ARRAYS

This section is a "brief" introduction to the idea of macro variable arrays. Let’s start with a definition.

As mentioned above, a macro variable array (an MVA or a macro array) is a list of macro variables with a
common prefix and numerical suffixes. There are many ways to create a macro array. The following are some
of the "most popular" ways for creating a macro array (for color convention check Appendix Appendix A -
code coloring guide).

Direct creation of an MVA can be simple assignment values to macro variables:
code: simple MVA

1 %let a1 = Macro;
2 %let a2 = Variables;
3 %let a3 = Array;

In this case, because of the prefix ("A"), we will say the macro array is "three element macro array a".

Another approach to create a macro array is to use data step and call symputx() routine:
code: data step approach

1 data _null_;
2 length val $ 9;
3 do val = "Next", "Macro", "Variables", "Array";
4 i+1;
5 call symputx(cats("b", i), val);
6 end;
7 run;

This time we are creating "four element macro array b".

Third approach is to use SAS Proc SQL’s into clause to create a macro array:
code: proc sql approach

1 data have;
2 input val $9.;
3 cards;
4 Proc
5 SQL
6 Macro
7 Variables
8 Array
9 ;

10 run;
11

12 proc SQL;
13 select val
14 into :c1-
15 from have;
16 quit;

The Proc SQL creates "five element macro array c".

Now when we execute:
code: preview

1 %put _user_;

we will see in the log that a dozen of macro variables were created:

the log
1 GLOBAL A1 Macro
2 GLOBAL A2 Variables
3 GLOBAL A3 Array
4 GLOBAL B1 Next
5 GLOBAL B2 Macro
6 GLOBAL B3 Variables
7 GLOBAL B4 Array
8 GLOBAL C1 Proc
9 GLOBAL C2 SQL

10 GLOBAL C3 Macro
11 GLOBAL C4 Variables
12 GLOBAL C5 Array
13 ...

The ". . ." indicates that there can be other macro variables created in the SAS session. In fact, since we are
using Proc SQL, there are a number of macro variables automatically created (not shown).

In a nut shell this is how MVAs can be created. The following articles provides good additional learning:
[Fehd 2003], [Carpenter 2017], [Renauldo 2018], [Horstman 2019], and of course the [Carpenter 2016]
book.

When an MVA has been created, to retrieve values from macro variables a technique called indirect referencing
is used. The indirect reference uses the following "multiple-ampersands" approach:

&&prefix&suffix

The method works as follows. Let’s assume we have our a1, a2, and a3 MVA. To print all three values in the
log the following macro loop does the job:

code: macro loop
1 %do i = 1 %to 3;
2 %put &&a&i;
3 %end;

(of course the loop is executed inside some macro!). In this case the prefix part is a and the suffix is i. The
code execution explanation is described below:

1. Value of i equals 1 which is less or equal to 3, loop starts the first iteration:
◽ SAS starts scanning the %put statement text and processes it;
● Two consecutive ampersands "&&" in the "&&a&i" code are recognized as a macro language triggers

and are replaced with single "&", text kept on stack is "&";
● The text "a" in the remaining "a&i" code, since it is not a macro language element, is added to the

stack, which is now "&a";
● The "&i" in the remaining "&i" code, is recognized as a macro variable i and its value is resolved, text

"1" is added to the stack, which is now "&a1";
● SAS gets to the end of the %put statement (the semicolon);
● SAS "sees" that text on the stack ("&a1") still has macrotriggers (ampersands) in it and starts pro-

cessing again.
3

◽ The "&a1" in the "&a1" code, is recognized as a macro variable a1 and its value is resolved, text
"Macro" is added to the stack, stack is now "Macro";
◽ SAS "sees" that there are no macrotriggers in the stack text and runs the %put statement;
◽ Value "Macro" is printed in the log.
◽ Value of i is increased by 1;

2. Value of i equals 2 which is less or equal to 3, loop starts the second iteration:
◽ SAS starts scanning the %put statement text and processes it;
● Two consecutive ampersands "&&" in the "&&a&i" code are recognized as a macro language triggers

and are replaced with single "&", text kept on stack is "&";
● The text "a" in the remaining "a&i" code, since it is not a macro language element, is added to the

stack, which is now "&a";
● The "&i" in the remaining "&i" code, is recognized as a macro variable i and its value is resolved, text

"2" is added to the stack, which is now "&a2";
● SAS gets to the end of the %put statement (the semicolon);
● SAS "sees" that text on the stack ("&a2") still has macrotriggers (ampersands) in it and starts pro-

cessing again.
◽ The "&a2" in the "&a2" code, is recognized as a macro variable a2 and its value is resolved, text

"Variables" is added to the stack, stack is now "Variables";
◽ SAS "sees" that there are no macrotriggers in the stack text and runs the %put statement;
◽ Value "Variables" is printed in the log.
◽ Value of i is increased by 1;

3. Value of i equals 3 which is less or equal to 3, loop starts the third iteration:
◽ SAS starts scanning the %put statement text and processes it;
● Two consecutive ampersands "&&" in the "&&a&i" code are recognized as a macro language triggers

and are replaced with single "&", text kept on stack is "&";
● The text "a" in the remaining "a&i" code, since it is not a macro language element, is added to the

stack, which is now "&a";
● The "&i" in the remaining "&i" code, is recognized as a macro variable i and its value is resolved, text

"3" is added to the stack, which is now "&a3";
● SAS gets to the end of the %put statement (the semicolon);
● SAS "sees" that text on the stack ("&a3") still has macrotriggers (ampersands) in it and starts pro-

cessing again.
◽ The "&a3" in the "&a3" code, is recognized as a macro variable a3 and its value is resolved, text

"Array" is added to the stack, stack is now "Array";
◽ SAS "sees" that there are no macrotriggers in the stack text and runs the %put statement;
◽ Value "Array" is printed in the log.
◽ Value of i is increased by 1;

4. Value of i equals 4 which is not less or equal to 3, loop stops. The end.

The result in the log is:
the log

1 Macro
2 Variables
3 Array

If you are "junior" SAS programmer and feel a bit overwhelmed by above explanation - do not be. All that
text can be boiled down to the following four rules:


 double ampersands (&&) are replaced with a single one (&),

 macro variables (&macVar) are resolved to the value (value),

 any non-macro text is kept "as is",

 repeat until there are no more macrotriggers.

When it goes to creation of MVAs, it can be seen that each of the above three methods has some "short-
comings". In the first one, it is obvious that the shortcoming is that it was created "by-hand". The second
requires jumping back from the macro language level to the 4GL data step. The third in addition to "4GL
back jumping" also requires a data set with input values. Further more, for inexperienced users the use of
"multiple-ampersands" approach to resolve macro variables data may be a bit "intimidating" or even ob-
scured1 and for experienced one it may cause "my-code-is-a-mess" thinking. For all that "little pains" the
SAS users community has solutions!

STANDING ON THE SHOULDERS OF GIANTS

The SAS community is a very active one, so in no time, solutions allowing to "improve" convenience of
work started to appear. Those of the highest value were articles presented by Ted Clay ([Clay 2004] and
[Clay 2006]) where the %array() and the %do_over() macros were presented. The purpose of the
%array() macro is to create MVAs in an easy and convenient way and the purpose of the %do_over()
macro is to allow practical looping over MVAs created by the %array(). For example:

code: Ted Clay’s approach
1 data Have;
2 input dayName :$12. value;
3 cards;
4 Monday 1
5 Tuesday 2
6 Wednesday 3
7 Thursday 4
8 Friday 5
9 Saturday 6

10 Sunday 7
11 ;
12 run;
13

14 %ARRAY(days, VALUES=Monday Wednesday Friday Sunday)
15

16 data Want;
17 set Have end=end;
18 where dayName in ( %DO_OVER(days, PHRASE="?"));
19 total + value;
20 if end then output;
21 run;

The %array() macro creates four global macro variables days1=Monday, days2=Wednesday,
days3=Friday, days4=Sunday, and additional (technical) variable daysN with value 4, which keeps
the information about the MVA length. The %do_over(), with help of daysN, executes internal loop over
days* and produce a text string:

"Monday" "Wednesday" "Friday" "Sunday"

which is used inside the where clause. Further more, the %array() macro also allows for the creation of
MVAs from data sets.

The section title "Standing on the shoulders of giants" (a metaphor which means "using the understanding
gained and work done by major thinkers working in the field before you in order to make progress") is fully
justified here since the first contact with those macros, used to solve problems on the SAS-L discussion list
and, in consequence, lecture of Ted Clay’s articles gave birth to the following thoughts in the authors head:

1. "This is brilliant!"

1Despite the "awkwardness" and "this-looks-to-hard-to-me" the author highly recommend putting effort and learn-
ing/understanding in depth how the process works.

2. "How does it work?!" and (after answering the question)
3. "Can its functionality be improved?"

Eventually, the development of the macroArray package was the result. One more remark, it is "giants",
not "giant", because the source code clearly states: "Authors: Ted Clay, David Katz"!

So, how was it with the macroArray package development in the first place? After reading through Clay’s
and Katz’s %array() code the first observation was that the code is not "pure macro code" 2 (to be more
precise, the part responsible for producing MVAs from data sets). According to the code information notes,
the last modification of the Clay’s and Katz’s %array()macro was in August 2006. At that point the famous
DOSUBL() function (see [Langston 2013], [McMullen 2020], or [Jablonski(1) 2023]) was not at their dis-
posal to utilize the "macro-function sandwich" technique. Fortunately, when I was taking my first attempts
to write my own version in January 2019 (see SAS-L post: "%array() macro and dosubl() function" - January
18, 2019, although do not use version of the code from that post, it is quite obsolete now) I was fortunate
to have the DOSUBL() at hand. My experience with the DOSUBL() led me to my first, and then several
consecutive, question.

So the first question was: can we have %array() working as "pure macro code" (even for creating macro
arrays from data sets)? The answer is yes.

The next observation was, macro arrays are "one based", i.e., the first element has always index 1. The second
question was: can an MVA created by %array() start indexing from an arbitrary non-negative integer (e.g.,
0, 2, 3, etc.)? Here, the answer is also yes.

Following idea was: we can create multiple MVAs from multiple variables from a data set. But can we also
create MVAs with lists of unique values? Yes.

When macro array is created its values still have to be called by &&prefix&suffix approach. Can
we make this call to the macro array value more "convenient", e.g., similar to the data step array, like:
(prefix[suffix])? Yes.

Can we use SAS functions to alter, modify, or generate MVA values? Yes.

When creating an MVA, can we have syntax similar to the one known form 4GL? Yes we can.

And one more, very important question. Can all those "yes we can" extension be added keeping backward
compatibility? Also, yes.

So, it is nice that all the answers were: yes we can! But now, our little internal "Bob the Builder" wants details.
Explanation to the questions stated above can be found in the next section where the macroArray package
is described.

THE MACROARRAY PACKAGE

In this section we are going to discuss the macroArray package, its components, capabilities, and possible
use cases.

The latest (and historical) version of the macroArray package, with its documentation, parameters de-
scription and examples, can be found at:

https://github.com/SASPAC/macroarray

2The "pure macro code" means it is a code which executes 100% on the macro processor level and does not produce any DATA
or PROC step code for the compilation.

[Note:] To start playing with any SAS Package, not only the macroArray package, the package has to be
first installed and loaded into the SAS session. The Appendix Appendix B - install the SPF and the package
provides detailed instruction of the process. Detailed information about SAS Packages idea can be found in
[Jablonski 2020], [Jablonski 2021], and [Jablonski(2) 2023].

Assuming we are ready to go, let’s see what the macroArray package has to offer.

At the moment the package contains 14 macros, which are: %appendArray(), %appendCell(),
%array(), %concatArrays(), %deleteMacArray(), %do_over(), %do_over2(),
%do_over3(), %make_do_over(), %mcHashTable(), %mcDictionary(), %QzipArrays(),
%zipArrays(), and %sortMacroArray().

Rewriting macroArray package’s documentation would be rather silly idea, especially since the documen-
tation is publicly available (see the github link above). Instead, we discover the macroArray package by
"getting our hands dirty", i.e., by going through a bunch of examples and use cases presenting package’s
features and capabilities.

The %array()macro
The first macro we discuss is the %array()3 macro. It allows us to create MVAs, either "from code" or from
a data set.

Let’s start with something simple. We want to create six element MVA "named" X with values from 101 to
106. For now, we will use the %put _user_; statement to print created macro variables.

code: %array() example No.1
1 %array(X[6] (101:106))
2

3 %put _user_;

In the log we can see the following:

the log
1 1 %array(X[6] (101:106))
2 NOTE:[ARRAY] 6 macrovariables created
3 2
4 3 %put _user_;
5 GLOBAL SYSLOADEDPACKAGES macroArray(X.Y.Z)
6 GLOBAL X1 101
7 GLOBAL X2 102
8 GLOBAL X3 103
9 GLOBAL X4 104

10 GLOBAL X5 105
11 GLOBAL X6 106
12 GLOBAL XHBOUND 6
13 GLOBAL XLBOUND 1
14 GLOBAL XN 6

Let’s take a look at what we see.
First, there is a note issued by %array() macro saying that six macro variables were created.
Second, syntax of the code looks "4GL-ish". Indeed when we like to create a data step array we write almost
exactly the same:

3To be clear, although the macros %array() and %do_over() were designed to replicate the functionality of Clay and Katz’s
macros, their source code was written from scratch. Only the names are the same.

code: 4GL array
1 data _null_;
2 array X[6] (101:106);
3 run;

This way of writing code is possible thanks to the DOSUBL() function working under the hood and giving us
a bit of "syntactic sugar" flavor which allows us to learn how to use the %array() macro faster.
Third, is a printout of user’s macro variables. We can see six global macro variables X1 to X6with values from
101 to 106 which "constitute" the macro array. Additionally, we can see macro variable XN which indicates
the number of macro array elements, similarly to the original Clay’s and Katz’s %array() macro. What is
new are two macro variables XLBOUND and XHBOUND which indicates lower and, respectively, higher index
of the macro array, in this example they are 1 and 6 respectively. We discuss XLBOUND and XHBOUND more
in a bit, but now let’s take a look at one more macro variable, namely the SYSLoadedPackages.

We are using the macroArray package which was loaded into the SAS session by the SAS Packages Frame-
work. The SYSLoadedPackages is special technical macro variable generated/updated by the framework
when a package is loaded into the SAS Session. The SYSLoadedPackages keeps information about all
SAS packages loaded in current SAS session. The value of the SYSLoadedPackages has the following
form: "package1(version) package2(version) ... packageN(version)". In the future
log printouts of user’s macro variables, for brevity, that one will be skipped.

[NOTE: About array name length] The naming convention for macro arrays is "SAS standard" with the respect
to the following restrictions: cannot be empty, cannot be longer than "32 − max(6, the number of
digits in the biggest suffix value)". In other words, if the maximum suffix is 123456789
then maximum length of a macro array name can be up to 23 (32 −max(6,9)) symbols4. The "6" is from
the "HBOUND" and "LBOUND" suffixes. There is one more "special case", when macro array’s name is single
underscore, but this case is explained separately and in details in the documentation.

Before further discussion, let’s take a look at the next example. We create the daysmacro array, with values:
Monday, Wednesday, Friday, and Sunday, like in the example from previous section. Since we can do
it on at least two ways, the first MVA will be named: days_A and the second: days_B.

code: %array() example No.2
1 %array(days_A[*] Monday Tuesday Wednesday Saturday, vnames=Y)
2

3 %array(days_B[4] $ 20 ("Monday" "Tuesday" "Wednesday" "Saturday"))
4

5 %put _user_;

The log shows the following:

the log - with "vnames=" parameter
1 1 %array(days_A[*] Monday Tuesday Wednesday Saturday, vnames=Y)
2 NOTE:[ARRAY] 4 macrovariables created
3 2
4 3 %array(days_B[4] $ 20 ("Monday" "Tuesday" "Wednesday" "Saturday"))
5 NOTE:[ARRAY] 4 macrovariables created
6 4
7 5 %put _user_;
8 GLOBAL DAYS_A1 Monday
9 GLOBAL DAYS_A2 Tuesday

10 GLOBAL DAYS_A3 Wednesday

4By "symbols" we mean the 26 letters of the Roman alphabet, ten digits from 0 to 9 and the underscore character.
8

11 GLOBAL DAYS_A4 Saturday
12 GLOBAL DAYS_AHBOUND 4
13 GLOBAL DAYS_ALBOUND 1
14 GLOBAL DAYS_AN 4
15 GLOBAL DAYS_B1 Monday
16 GLOBAL DAYS_B2 Tuesday
17 GLOBAL DAYS_B3 Wednesday
18 GLOBAL DAYS_B4 Saturday
19 GLOBAL DAYS_BHBOUND 4
20 GLOBAL DAYS_BLBOUND 1
21 GLOBAL DAYS_BN 4

We can see in both cases that a macro array was generated, but in each case a different approach was used.
In the first approach we use the vnames=Y parameter and the fact that values: Monday, Wednesday,
Friday, and Sunday can be used as data step variables names. The vnames=Y parameter instructs the
%array() macro to take variable’s names as MVA’s values instead "real" values stored in variables (the
default behavior). If we do not use the vnames= parameter or set its value to anything other then Y, the
result is:

the log - no "vnames=" parameter
1 1 %array(days_A[*] Monday Tuesday Wednesday Saturday)
2 NOTE:[ARRAY] 4 macrovariables created
3 2
4 3 %put _user_
5 GLOBAL DAYS_A1 .
6 GLOBAL DAYS_A2 .
7 GLOBAL DAYS_A3 .
8 GLOBAL DAYS_A4 .
9 GLOBAL DAYS_AHBOUND 4

10 GLOBAL DAYS_ALBOUND 1
11 GLOBAL DAYS_AN 4

The reason we get "periods" is because (according to 4GL rules) the array under the hood is considered to
be numeric. Bottom line is, if values (e.g., week days names) you want to use for MVA values "satisfy" SAS
variables naming convention you can do it with vnames=Y.
In the second approach by writing "[4] $ 20" we are declaring a character array of size four with initial
values which are then used to generate the MVA. This approach does not have "limitations" of the first one.
Furthermore since values are provided in quotes it can be very handy way of "masking" special characters
like ampersand, percent, semicolon or comma in macro array values, for example:

the log - "specials"
1 1 %array(days_B[4] $ 20 (’&Monday.’ ’%Tuesday()’ "Wed;nes;day;" "Sa,tur,day")
2 NOTE:[ARRAY] 4 macrovariables created
3 2
4 3 %put _user_
5 GLOBAL DAYS_B1 &Monday.
6 GLOBAL DAYS_B2 %Tuesday()
7 GLOBAL DAYS_B3 Wed;nes;day;
8 GLOBAL DAYS_B4 Sa,tur,day
9 GLOBAL DAYS_BHBOUND 4

10 GLOBAL DAYS_BLBOUND 1
11 GLOBAL DAYS_BN 4

Obviously the "Wed;nes;day;" text can be replaced with something like: "proc print
data=sashelp.class; var name age; run;".

9



B. Jabłoński - Macro Variable Arrays Made Easy with macroArray SAS Package

In the examples above values of the MVA were a "static text", but this does not have to be the case. The
%array() macro can be extended with the use of SAS functions that can be used to generate macro array
values. A dedicated parameter, function=, can be used to assign the array’s values. The function=
parameter expects a SAS 4GL function, composition of functions, or an expression using functions. If we
would like to use array’s index in an expression the index value is kept in the _i_ variable5. Important thing
to remember is that if the type of an array it numeric, then the value returned by the expression infunction=
parameter must "align" (be numeric too). Let’s look at log output for some examples:

the log - with "function=" parameter
1 1 /* three constant expression, equiv. of %array(e[1:3] $ (3*"A") ) */
2 2 %array(e[1:3] $, function="A" )
3 NOTE:[ARRAY] 3 macrovariables created
4 3
5 4 /* first six integer powers of 2 */
6 5 %array(f[0:5], function=(2**_i_) )
7 NOTE:[ARRAY] 6 macrovariables created
8 6
9 7 /* five random numbers form uniform distribution on (0,1) interval */

10 8 %array(g[0:4], function=ranuni(123) )
11 NOTE:[ARRAY] 5 macrovariables created
12 9
13 10 /* a formated list of twelve months */
14 11 %array(h[0:11] $ 11, function=put(intnx("MONTH","19may2024"d, _i_),yymmd.))
15 NOTE:[ARRAY] 12 macrovariables created
16 12
17 13 /* Fibonacci sequence, first ten elements */
18 14 %array(i[10] (10*0)
19 15 ,function=ifn(_i_<2, 1, sum(i[max(_i_-2,1)], i[max(_i_-1,2)])))
20 NOTE:[ARRAY] 10 macrovariables created
21 16
22 17 %put _user_;
23 GLOBAL E1 A
24 GLOBAL E2 A
25 GLOBAL E3 A
26 GLOBAL EHBOUND 3
27 GLOBAL ELBOUND 1
28 GLOBAL EN 3
29

30 GLOBAL F0 1
31 GLOBAL F1 2
32 GLOBAL F2 4
33 GLOBAL F3 8
34 GLOBAL F4 16
35 GLOBAL F5 32
36 GLOBAL FHBOUND 5
37 GLOBAL FLBOUND 0
38 GLOBAL FN 6
39

40 GLOBAL G0 0.7503960881
41 GLOBAL G1 0.3209120251
42 GLOBAL G2 0.178389649
43 GLOBAL G3 0.9060333813
44 GLOBAL G4 0.3571170775
45 GLOBAL GHBOUND 4
46 GLOBAL GLBOUND 0
47 GLOBAL GN 5

5The _i_ variable is used internaly inside the %array() macro processing.
10

48

49 GLOBAL H0 2024-05
50 GLOBAL H1 2024-06
51 GLOBAL H2 2024-07
52 GLOBAL H3 2024-08
53 GLOBAL H4 2024-09
54 GLOBAL H5 2024-10
55 GLOBAL H6 2024-11
56 GLOBAL H7 2024-12
57 GLOBAL H8 2025-01
58 GLOBAL H9 2025-02
59 GLOBAL H10 2025-03
60 GLOBAL H11 2025-04
61 GLOBAL HHBOUND 11
62 GLOBAL HLBOUND 0
63 GLOBAL HN 12
64

65 GLOBAL I1 1
66 GLOBAL I2 1
67 GLOBAL I3 2
68 GLOBAL I4 3
69 GLOBAL I5 5
70 GLOBAL I6 8
71 GLOBAL I7 13
72 GLOBAL I8 21
73 GLOBAL I9 34
74 GLOBAL I10 55
75 GLOBAL IHBOUND 10
76 GLOBAL ILBOUND 1
77 GLOBAL IN 10

The function= parameter with all its "functionality" is not all that is. There are two more (optional) coun-
terpart parameters: before= and after=. The before= expects a function or an expression to be added
before looping through the array. The after= expects a function or an expression to be added after looping
through the array. If we want to provide a series of statements to before= or after= they should be sep-
arated by semicolons. The following example generates macro array j with sorted random numbers in range
from zero to one, generated by the rand() function from uniform distribution. The seed for the rand()
function is set with the help of before= parameter. Sorting, change of the first and the last value, and a
"debugging printout" are done with after= parameter:

code: %array() example No.3
1 %array(j[10]
2 ,function = round(rand(’Uniform’,0,1),0.001)
3 ,before = call streaminit(42)
4 ,after = call sortn(of j[*]); j[1]=0; j[10]=1; put _all_
5 )
6

7 %put _user_

The log presents the following content:

the log - with "before=" and "after=" parameters
1 1 %array(j[10]
2 2 ,function = round(rand(’Uniform’,0,1),0.001)
3 3 ,before = call streaminit(42)
4 4 ,after = call sortn(of j[*]); j[1]=0; j[10]=1; put _all_
5 5 )

11

6 j1=0 j2=0.137 j3=0.219 j4=0.513 j5=0.66
7 j6=0.728 j7=0.747 j8=0.753 j9=0.901 j10=1
8 _I_=11 _ERROR_=0 _N_=1
9 NOTE:[ARRAY] 10 macrovariables created

10 6
11 7 %put _user_;
12 GLOBAL J1 0
13 GLOBAL J2 0.137
14 GLOBAL J3 0.219
15 GLOBAL J4 0.513
16 GLOBAL J5 0.66
17 GLOBAL J6 0.728
18 GLOBAL J7 0.747
19 GLOBAL J8 0.753
20 GLOBAL J9 0.901
21 GLOBAL J10 1
22 GLOBAL JHBOUND 10
23 GLOBAL JLBOUND 1
24 GLOBAL JN 10

So at this point an MVA created by macroArray package has all properties provided by the Clay’s and Katz’s
%array() macro, so we can also use it with their version of %do_over().

Next extension of the %array() macro is the possibility to index "hand coded" macro arrays not only start-
ing from one but from any non-negative integer (e.g., 0, 2, 3, etc.), for example:

code: %array() example No.4
1 %array(k[5:7] $ ("fifth" "sixth" "seventh"))
2

3 %put _user_;

In the log we see:

the log - with "arbitrary index"
1 1 %array(k[5:7] $ ("fifth" "sixth" "seventh"))
2 NOTE:[ARRAY] 3 macrovariables created
3 2
4 3 %put _user_;
5 GLOBAL K5 fifth
6 GLOBAL K6 sixth
7 GLOBAL K7 seventh
8 GLOBAL KHBOUND 7
9 GLOBAL KLBOUND 5

10 GLOBAL KN 3

In case we decide to start indexing from a negative integer (e.g. -3) then the index range value in the variable
_i_ is used "as is", i.e., starts from provided negative value, but... and this has to be highlighted(!), the index
range for macro array is automatically shifted to start at zero! For example if we want to get powers of two
with exponents from -3 to 3 in macro array we can do the following:

code: %array() example No.5
1 %array(l[-3:3], function = 2**_i_)
2

3 %put _user_;

12

In this case the log shows:

the log - with "negative index"
1 1 %array(l[-3:3], function = 2**_i_)
2 NOTE:[ARRAY] 7 macrovariables created
3 2
4 3 %put _user_;
5 GLOBAL L0 0.125
6 GLOBAL L1 0.25
7 GLOBAL L2 0.5
8 GLOBAL L3 1
9 GLOBAL L4 2

10 GLOBAL L5 4
11 GLOBAL L6 8
12 GLOBAL LHBOUND 6
13 GLOBAL LLBOUND 0
14 GLOBAL LN 7

So the values assigned are 2−3, 2−2, . . ., 22, 23 but the suffixes goes from zero to six.
The reason for implementing such behavior is based on the following considerations. The only characters al-
lowed for macro variables names are letters, digits and underscore("_"). What symbol could be used to indi-
cate minus sign in negative index? It cannot be a number or a letter, the first are already reserved for suffix part
the second for prefix. The only "natural candidate" which pops up is the underscore. But if we decide to use
underscore as an indicator of negative value, then from looking at the following macro array M_1=2, M_2=1
you cannot tell which of those two examples: %array(m[-2:-1] (1 2)) and %array(m_[1:2] (2
1)), was used to produce the MVA. In other words, we cannot tell if it is array M_ with index 1, or if it is array
M with index -1.

But macro arrays created from data sets are indexed from one! Oh! Right! We didn’t mention this feature yet!
Macro variable arrays can be created from a data set.

The %array() macro allows us to create a macro array from a variable in a data set. Even better, multiple
macro arrays can be created based on different variables from that data set. Additionally, it is possible to
request a macro array with only distinct values extracted from a variable. To create MVA from a variable in a
data set two parameters should be used. The first one, ds=, points to a data set, the second, vars=, points
to variables. The ds= parameter takes precedence over other parameters, i.e., if both ds= is not empty
(points to a data set) and "by code"6 array definition is provided then the ds= dedicated code is executed.
Data set in the ds= can be provided with data set options in parentheses but the data set can be (like in the
"Highlander" movie) only one. On the other hand, the vars= parameter allows for multiple values, this will
be discussed next, but now let’s take a look at the following example:

code: %array() example No.6
1 %array(ds=sashelp.class(where=(age=13)), vars=name)
2

3 %put _user_;

The condition in the where= data set option selects three observations from sashelp.class data set
and values of the name variable are taken for MVA values, the name of the variable is used for MVA name.
Log shows the following result:

6A "by code" array is an array declared within the macro call, the one we discussed earlier.
13

the log - with "ds=" and "vars=" parameter
1 1 %array(ds=sashelp.class(where=(age=13)), vars=name)
2 NOTE:[ARRAY] 3 macrovariables created
3 2
4 3 %put _user_;
5 GLOBAL NAME1 Alice
6 GLOBAL NAME2 Barbara
7 GLOBAL NAME3 Jeffrey
8 GLOBAL NAMEHBOUND 3
9 GLOBAL NAMELBOUND 1

10 GLOBAL NAMEN 3

If we decide to create multiple MVAs we simply list variables we want to use separated by a space, for example
like this:

code: %array() example No.7
1 %array(ds=sashelp.class(where=(age=13)), vars=name height weight)
2

3 %put _user_;

Log from execution is the following:

the log - ith multiple variables
1 1 %array(ds=sashelp.class(where=(age=13)), vars=name height weight)
2 NOTE:[ARRAY] 9 macrovariables created
3 2
4 3 %put _user_;
5 GLOBAL HEIGHT1 56.5
6 GLOBAL HEIGHT2 65.3
7 GLOBAL HEIGHT3 62.5
8 GLOBAL HEIGHTHBOUND 3
9 GLOBAL HEIGHTLBOUND 1

10 GLOBAL HEIGHTN 3
11

12 GLOBAL NAME1 Alice
13 GLOBAL NAME2 Barbara
14 GLOBAL NAME3 Jeffrey
15 GLOBAL NAMEHBOUND 3
16 GLOBAL NAMELBOUND 1
17 GLOBAL NAMEN 3
18

19 GLOBAL WEIGHT1 84
20 GLOBAL WEIGHT2 98
21 GLOBAL WEIGHT3 84
22 GLOBAL WEIGHTHBOUND 3
23 GLOBAL WEIGHTLBOUND 1
24 GLOBAL WEIGHTN 3

First we observe three brand new MVAs created, the second observation takes us to the log note printed by
the macro. The note reports total number of macro variables created.

The vars= parameter has two more features. The first, is a possibility to select only unique values from a
variable. It can be done by adding a vertical bar character ("|") directly after the variable name in the list, in
other words "no space in-between", like in this example:

14

code: %array() example No.8
1 %array(ds=sashelp.class, vars=age|)
2

3 %put _user_;

Log shows the following:

the log - with unique values
1 1 %array(ds=sashelp.class, vars=age|)
2 NOTE:[ARRAY] 6 macrovariables created
3 2
4 3 %put _user_;
5 GLOBAL AGE1 15
6 GLOBAL AGE2 12
7 GLOBAL AGE3 16
8 GLOBAL AGE4 13
9 GLOBAL AGE5 14

10 GLOBAL AGE6 11
11 GLOBAL AGEHBOUND 6
12 GLOBAL AGELBOUND 1
13 GLOBAL AGEN 6

We can see six unique values selected. Order of returned values, as can be seen, is random. In general: if
there is "|" inserted behind a variable name unique values are selected, and if there is nothing or the pound
sign ("#") behind a variable name then values from all observations are selected.

A curious reader instantly asks: Why additional symbol (the pound sign)? The vertical bar(|) and "nothing"
perfectly good distinguish "select distinct vs. select all" values. Well, it would be enough but only in situations
when we want MVA name to be the same as the data set variable’s name. And this is the second feature the
vars= parameter has. We can use values from a variable but we do not need to use its name, we can change
MVA’s name by setting an alias. The syntax is:

variable1<| or #<alias1>> <variable2<| or #<alias2>>> ... <variableN<|
or #<aliasN>>>

The <...> indicates optional element. Basically, the variableA|aliasA means: select distinct val-
ues from variableA into MVA named aliasA, and the variableB#aliasB means: select all val-
ues from variableB into MVA named aliasB. The best is to look at the next example in which we use
sashelp.class and create macro array N with all values from variable name and macro array A with
unique values from variable age:

code: %array() example No.9
1 %array(ds=sashelp.class, vars=name#N age|A)
2

3 %put _user_;

And, as we would expect, in the log we see 25 variables created:

the log - with "vars=" parameter in full
1 1 %array(ds=sashelp.class, vars=name#N age|A)
2 NOTE:[ARRAY] 25 macrovariables created
3 2
4 3 %put _user_;
5 GLOBAL A1 15
6 GLOBAL A2 12
7 GLOBAL A3 16

15

8 GLOBAL A4 13
9 GLOBAL A5 14

10 GLOBAL A6 11
11 GLOBAL AHBOUND 6
12 GLOBAL ALBOUND 1
13 GLOBAL AN 6
14 GLOBAL N1 Alfred
15 GLOBAL N2 Alice
16 GLOBAL N3 Barbara
17 ...
18 GLOBAL N17 Ronald
19 GLOBAL N18 Thomas
20 GLOBAL N19 William
21 GLOBAL NHBOUND 19
22 GLOBAL NLBOUND 1
23 GLOBAL NN 19

We all know that in the macro language world every value is a text so no quotes are needed. But even despite
that, from time to time there is a need that value must be presented in quotes (double or single). That is
why one more parameter, q=, which indicate if quotes should be added, is available. The parameter can
take two values, 2 to indicate quoting with double quotes and 1 to indicate quoting with apostrophes (single
quotes). Parameter works both with "coded" macro arrays and with "from data set" arrays. Log for two simple
examples looks like this:

the log - with "q=" parameter
1 1 %array(Letters[5] $ 3, function = byte(rank(’A’)+_I_-1) , q=2)
2 NOTE:[ARRAY] 5 macrovariables created
3 2
4 3 %array(ds=sashelp.class, vars=name, q=1)
5 NOTE:[ARRAY] 19 macrovariables created
6 4
7 5 %put _user_;
8 GLOBAL LETTERS1 "A"
9 GLOBAL LETTERS2 "B"

10 GLOBAL LETTERS3 "C"
11 GLOBAL LETTERS4 "D"
12 GLOBAL LETTERS5 "E"
13 GLOBAL LETTERSHBOUND 5
14 GLOBAL LETTERSLBOUND 1
15 GLOBAL LETTERSN 5
16 GLOBAL NAME1 ’Alfred’
17 GLOBAL NAME2 ’Alice’
18 GLOBAL NAME3 ’Barbara’
19 ...
20 GLOBAL NAME17 ’Ronald’
21 GLOBAL NAME18 ’Thomas’
22 GLOBAL NAME19 ’William’
23 GLOBAL NAMEHBOUND 19
24 GLOBAL NAMELBOUND 1
25 GLOBAL NAMEN 19

[Note:] If you are working with BIG macro arrays do not forget to verify your session setting for macro memory
limits. Run:

code: session setting for macro memory
1 proc options group = macro;
2 run;

to verify the following options:


 MEXECSIZE - specifies the maximum macro size that can be executed in memory.

 MSYMTABMAX - specifies the maximum amount of memory available to the macro variable symbol table

or tables.

 MVARSIZE - specifies the maximum size for a macro variable that is stored in memory.

In the final part of this subsection we discuss one more feature, unique to the macroArray package’s
implementation of the %array() macro. Up to now, to display macro array’s elements in the log we use the
%put _user_; trick. But what to do when we want to use MVA’s elements separately, e.g., in a loop of some
sort? As we already know, the "structure" of variables in a macro array has the form: [prefix][suffix]
and when we want to call their values we have to use indirect reference. To call elements of the macro array
ARR in the example below we use the &&prefix&suffix construction in the %DO-LOOP:

code: %array() example No.10
1 %macro someMacro();
2 %array(ARR[1:3] $ ("first" "second" "third"))
3

4 %do i=1 %to 3;
5 %put for &=i. value is &&ARR&i ;
6 %end;
7 %mend someMacro;
8

9 %someMacro()

In the log, printout from the %someMacro() execution has the following form:

the log - getting values
1 1 %someMacro()
2 NOTE:[ARRAY] 3 macrovariables created
3 for I=1 value is first
4 for I=2 value is second
5 for I=3 value is third

To make macro arrays behavior even more "4GL arrays" alike the macarray=<Y,N> parameter was intro-
duced. The purpose of the macarray=, when set to Y (the default is N), is to generate a new macro named
the same as the MVA name. The new macro that is created allows us to call to macro arrays elements in a
similar fashion we call array elements in the data step, with the difference being that now the percent symbol
(%) precedes the reference. The example above can be modified the following way:

code: %array() example No.11
1 %macro someMacro2();
2 %array(ARR[1:3] $ ("first" "second" "third"), macarray=Y)
3

4 %do i=1 %to 3;
5 %put for &=i. value is %ARR(&i) ;
6 %end;
7 %mend someMacro2;
8

9 %someMacro2()

17

And in the log, printout from the %someMacro2() execution looks exactly the same:

the log - getting values with "macarray=" parameter
1 1 %someMacro2()
2 NOTE:[ARRAY] 3 macrovariables created
3 for I=1 value is first
4 for I=2 value is second
5 for I=3 value is third

Clearly the %ARR(&i) is more "array like" looking code than the &&ARR&i is, isn’t it? Plus the added benefit
is you don’t have to remember how many & are needed.

The parameter works with data set generated macro arrays too. And when used with lower and high bound
variables it works even better, like in this example:

code: %array() example No.12
1 %macro someMacro3();
2 %array(ds=sashelp.class, vars=age|ARR, macarray=Y)
3

4 %do i=&ARRLBOUND. %to &ARRHBOUND.;
5 %put for &=i. value is %ARR(&i) ;
6 %end;
7 %mend someMacro3;

Log printout:

the log - with "macarray=" and "ds=" parameters
1 1 %someMacro3()
2 NOTE:[ARRAY] 6 macrovariables created
3 for I=1 value is 15
4 for I=2 value is 12
5 for I=3 value is 16
6 for I=4 value is 13
7 for I=5 value is 14
8 for I=6 value is 11
9 2 %put _user_;

10 GLOBAL ARR1 15
11 GLOBAL ARR2 12
12 GLOBAL ARR3 16
13 GLOBAL ARR4 13
14 GLOBAL ARR5 14
15 GLOBAL ARR6 11
16 GLOBAL ARRHBOUND 6
17 GLOBAL ARRLBOUND 1
18 GLOBAL ARRN 6

The "macro-array" macro created when the macarray= parameter is in use has one more convenient fea-
ture. It allows us to reassign values of existing macro variables. The process is very straightforward, all we
need to do is to set the "macro-array" macro’s second argument to "i" (like "insert"). Here is the example:

code: %array() example No.13
1 %array(ABC[3] (1:3), macarray=Y);
2

3 %put _user_;
18

4 %let %ABC(2,i) = 99999;
5 %put _user_;

With log print showing:

the log - modifying macro array values
1 1 %array(ABC[3] (1:3), macarray=Y);
2 NOTE:[ARRAY] 3 macrovariables created
3 2
4 3 %put _user_;
5 GLOBAL ABC1 1
6 GLOBAL ABC2 2
7 GLOBAL ABC3 3
8 GLOBAL ABCHBOUND 3
9 GLOBAL ABCLBOUND 1

10 GLOBAL ABCN 3
11 4 %let %ABC(2,i) = 99999;
12 5 %put _user_;
13 GLOBAL ABC1 1
14 GLOBAL ABC2 99999
15 GLOBAL ABC3 3
16 GLOBAL ABCHBOUND 3
17 GLOBAL ABCLBOUND 1
18 GLOBAL ABCN 3

The macarray= parameter provides one more useful feature. If its value is set to M ("make") then for a
given array name the macro symbols table is scanned for all macro variables with a prefix like the array name
and numeric suffixes. From all macro variables of that form found, the minimum and the maximum index is
determined and all non-existing sequential global macro variables between the minimum and maximum index
values are created. Also a macro is generated, in the same fashion as one is generated for the macarray=Y
value. Here is log printout from a small example:

the log - with macarray=M
1 1 %let ARR1=1;
2 2 %let ARR3=3;
3 3 %let ARR5=5;
4 4
5 5 %put _user_;
6 GLOBAL ARR1 1
7 GLOBAL ARR3 3
8 GLOBAL ARR5 5
9 6 %array(ARR,macarray=m)

10 NOTE:[ARRAY] 5 macrovariables created
11 7 %put _user_;
12 GLOBAL ARR1 1
13 GLOBAL ARR2
14 GLOBAL ARR3 3
15 GLOBAL ARR4
16 GLOBAL ARR5 5
17 GLOBAL ARRHBOUND 5
18 GLOBAL ARRLBOUND 1
19 GLOBAL ARRN 5

This ends the discussion about features and capabilities of the %array() macro.

The %deleteMacArray()macro
Purpose of the %deleteMacArray()macro is very simple, it is a "house keeping" macro. When macro ar-
ray is created with help of the %array()macro the macro variables are set as global, even if the %array()
macro execution creates an MVA on seventeenth, or forty-second level of macro nesting depth. This may,
when less attention is given, produce overwrite errors. To minimize probability of such error the best ap-
proach is to delete macro array as soon as it is no longer needed. The "possible use case" code could be
something like the following one:

code: clean after work
1 %macro someMacroX();
2 /* declare someMacroArray */
3 %array(<definition of someMacroArray>, macarray=Y)
4 /* code using someMacroArray */
5 %do i=&someMacroArrayLBOUND. %to &someMacroArrayHBOUND.;
6 /* <... call to %someMacroArray(&i) ...> */
7 %end;
8 /* delete someMacroArray */
9 %deleteMacArray(someMacroArray, macarray=Y)

10 %mend someMacroX;

The second argument of the %deleteMacArray() macro, the macarray=<Y,N>, indicates should a
macro associated with macro array be deleted too. Default is N, meaning "no".

If we specify a macro array to delete and there isn’t one an error in the log is produced. If we indicate
macarray=Y and there is no macro associated with the MVA the %deleteMacArray()macro still works
and deletes the macro array specified in the first argument.

The %sortMacroArray()macro
Another "house keeping" macro is the %sortMacroArray() macro. As the name suggests the macro
allows to sort macro array’s values, for example like this case:

code: sorting macro array
1 %array(n[6] $ 3 ("C33" "B22" "A11" "A01" "A02" "X42"))
2

3 %put _user_;
4

5 %sortMacroArray(n)
6

7 %put _user_;

Log shows the following:

the log - with "macarray=" and "ds=" parameters
1 1 %array(n[6] $ 3 ("C33" "B22" "A11" "A1" "A2" "X42"))
2 NOTE:[ARRAY] 6 macrovariables created
3 2
4 3 %put _user_;
5 GLOBAL N1 C33
6 GLOBAL N2 B22
7 GLOBAL N3 A11
8 GLOBAL N4 A1
9 GLOBAL N5 A2

10 GLOBAL N6 X42
20

11 GLOBAL NHBOUND 6
12 GLOBAL NLBOUND 1
13 GLOBAL NN 6
14 4
15 5 %sortMacroArray(n)
16 6
17 7 %put _user_;
18 GLOBAL N1 A1
19 GLOBAL N2 A2
20 GLOBAL N3 A11
21 GLOBAL N4 B22
22 GLOBAL N5 C33
23 GLOBAL N6 X42
24 GLOBAL NHBOUND 6
25 GLOBAL NLBOUND 1
26 GLOBAL NN 6

Though macro works as "pure macro code", internally theSORT procedure does the job, the sorting sequence
can be modified by the sortseq= parameter. Its default value is LINGUISTIC(NUMERIC_COLLATION
= ON) which sort data that contain numbers in the values as if they were numbers (e.g., "A2" before "A10").

Additionally, the outSet= macro parameter allows to store sorted values with the index "after sorting" in a
data set pointed by the parameter. An example of sorting the MVA n from the previous example and saving
its data in the sortedData data set looks like this:

code: sorting macro array and storing sorted data
1 %sortMacroArray(n, outSet=sortedData)

Since under the hood the SORT procedure is used, the macro has one limitation, maximum length of values
sorted is only 32767 bytes.

Macros %appendArray() and %concatArrays()
Two more macros that "organize macro arrays world" are %appendArray() and %concatArrays().
The first macro (in fact a wrapper over the second) allows to append one macro array’s values at the end of
another macro array. The second macro does the same as the first one does, but also deletes the second
array’s elements after appending. The log illustrates this situation the best:

the log - appending one to another
1 1 %array(o[2:4] $ 1 ("A" "B" "C"))
2 NOTE:[ARRAY] 3 macrovariables created
3 2
4 3 %array(p[3] (1 2 3))
5 NOTE:[ARRAY] 3 macrovariables created
6 4
7 5 %put _user_;
8 GLOBAL O2 A
9 GLOBAL O3 B

10 GLOBAL O4 C
11 GLOBAL OHBOUND 4
12 GLOBAL OLBOUND 2
13 GLOBAL ON 3
14 GLOBAL P1 1
15 GLOBAL P2 2
16 GLOBAL P3 3
17 GLOBAL PHBOUND 3
18 GLOBAL PLBOUND 1
19 GLOBAL PN 3

20 6 %appendArray(o, p);
21 7 %put _user_;
22 GLOBAL O2 A
23 GLOBAL O3 B
24 GLOBAL O4 C
25 GLOBAL O5 1
26 GLOBAL O6 2
27 GLOBAL O7 3
28 GLOBAL OHBOUND 7
29 GLOBAL OLBOUND 2
30 GLOBAL ON 6
31 GLOBAL P1 1
32 GLOBAL P2 2
33 GLOBAL P3 3
34 GLOBAL PHBOUND 3
35 GLOBAL PLBOUND 1
36 GLOBAL PN 3

And the second one:

the log - concatenating two
1 1 %array(q[2:4] $ 1 ("E" "F" "G"))
2 NOTE:[ARRAY] 3 macrovariables created
3 2
4 3 %array(r[3] (4 5 6))
5 NOTE:[ARRAY] 3 macrovariables created
6 4
7 5 %put _user_;
8 GLOBAL Q2 E
9 GLOBAL Q3 F

10 GLOBAL Q4 G
11 GLOBAL QHBOUND 4
12 GLOBAL QLBOUND 2
13 GLOBAL QN 3
14 GLOBAL R1 4
15 GLOBAL R2 5
16 GLOBAL R3 6
17 GLOBAL RHBOUND 3
18 GLOBAL RLBOUND 1
19 GLOBAL RN 3
20 6 %concatArrays(q, r);
21 7 %put _user_;
22 GLOBAL Q2 E
23 GLOBAL Q3 F
24 GLOBAL Q4 G
25 GLOBAL Q5 4
26 GLOBAL Q6 5
27 GLOBAL Q7 6
28 GLOBAL QHBOUND 7
29 GLOBAL QLBOUND 2
30 GLOBAL QN 6

In both cases we can also notice how indexes increase is automatically maintained.

The %do_over()macro
We already saw how the macarray= parameter of the %array() macro makes do-looping over macro

array elements simpler. A macro we discuss in this subsection, the %do_over() macro, is a helper macro
for the looping process itself. It allows for the looping over a macro array elements without the necessity of
"putting" the do-loop inside an extra/external macro. As the result the %do_over() generates plain text
which is then "used" (of course the result depends on the context into which the text is pushed in the code,
see the next two examples). Also when executed, the %do_over() macro assumes that the %array()
macro was run with the macarray= parameter set to Y.

For example the following:

code: silly %do_over() looping
1 %array(ds=sashelp.class(where=(age=12)), vars=name, q=2, macarray=Y)
2

3 %do_over(name)
4

5 %put _user_;

will rise an error message in the log:

the log - error
1 1 %array(ds=sashelp.class(where=(age=12)), vars=name, q=2, macarray=Y)
2 NOTE:[ARRAY] 5 macrovariables created
3 2
4 3 %do_over(name)
5 NOTE: Line generated by the macro function "UNQUOTE".
6 1 "James"
7 -------
8 180
9 ERROR 180-322: Statement is not valid or it is used out of proper order.

10

11 4
12 5 %put _user_;
13 GLOBAL NAME1 "James"
14 GLOBAL NAME2 "Jane"
15 GLOBAL NAME3 "John"
16 GLOBAL NAME4 "Louise"
17 GLOBAL NAME5 "Robert"
18 GLOBAL NAMEHBOUND 5
19 GLOBAL NAMELBOUND 1
20 GLOBAL NAMEN 5

because SAS does not know how to interpret the »"James" "Jane" ... "Robert"« plain text string
in the open code.

But when we add for example the %PUT statement or place the %do_over() result in proper context (e.g.
inside a DATA STEP code) it works as intended:

code: wise %do_over() looping
1 %array(ds=sashelp.class(where=(age=12)), vars=name, q=2, macarray=Y)
2

3 %put %do_over(name);
4

5 data test;
6 set sashelp.class;
7 where name in (%do_over(name));
8 run;

which can be confirmed in log printout:

the log - good looping
1 1 %array(ds=sashelp.class(where=(age=12)), vars=name, q=2, macarray=Y)
2 NOTE:[ARRAY] 5 macrovariables created
3 2
4 3 %put %do_over(name);
5 "James" "Jane" "John" "Louise" "Robert"
6 4
7 5 data test;
8 6 set sashelp.class;
9 7 where name in (%do_over(name));

10 8 run;
11

12 NOTE: There were 5 observations read from the data set SASHELP.CLASS.
13 WHERE name in (’James’, ’Jane’, ’John’, ’Louise’, ’Robert’);
14 NOTE: The data set WORK.TEST has 5 observations and 5 variables.

As we can see from the examples, when the %do_over() is executed in its most basic form, i.e., with only a
macro array name as the array parameter value, the result is a text string listing all elements of the macro
array separated by space character.

The basic form, which uses only a macro array name, has a lot of practical use cases but, when we use addi-
tional key-value parameters then the %do_over() reveals its full potential. The first key-value parameter is
the between= parameter, it provides a "separator" string which is inserted between consecutive iterations.
The default is space character.

The following example uses the »" "« text (double quote, space, and double quote) for the between= and
does not use the q=2 in the MVA definition, plus the %do_over() is "wrapped" in double quotes:

code: %do_over() looping with "between="
1 %array(ds=sashelp.class(where=(age=12)), vars=name, macarray=Y)
2

3 %put "%do_over(name, between=" ")";

Log shows the following:

the log - "between="
1 1 %array(ds=sashelp.class(where=(age=12)), vars=name, macarray=Y)
2 NOTE:[ARRAY] 5 macrovariables created
3 2
4 3 %put "%do_over(name, between=" ")" ;
5 "James" "Jane" "John" "Louise" "Robert"

We can observe that this and the previous example produce the same result.

Next key-value parameter is the phrase=, this parameter keeps the text of a phrase which is executed
in each iteration of the %do_over() (see examples). The default value of the parameter, which results
in printing all elements of a macro array, is the following phrase of text: "%NRSTR%&array(&_i_.)".
The macro variable &_i_. is an implicit internal iterator of the %do_over() macro, the &_i_. can be
used inside phrases you want to loop by the %do_over() macro7. There is a requirement(!) for phrases,
when calling to &_i_. or "macarray=Y" generated macros, that they have to be "enveloped" inside the

7The decission to use &_i_. instead &i. as the iterator was dictated by desire to keep similarity with 4GL’s "DO OVER Array;"
statement and its default iterator _i_

%NRSTR macro quoting function! Of course when the %NRSTR is used its masking rules, e.g., for brackets or
apostrophes, has to be kept. So in practice, when we are thinking: "phrase=<...code...>" it is better
to think: "phrase=%NRSTR(<...code...>)".

To make it clearer, let’s look at some examples. In the first one we want to split a single data set into separate
data sets based on a variable’s values:

code: using "phrase=" parameter, ex. 1
1 %array(ds=sashelp.class, vars=age|, macarray=Y)
2

3 options mprint;
4 data
5 %do_over(age, phrase=%nrstr(data_%age(&_i_.)))
6 ;
7 set sashelp.class;
8

9 select(age);
10 %do_over(age, phrase=%nrstr(
11 when (%age(&_i_.)) output data_%age(&_i_.);
12 ))
13 otherwise put "unknown Age value!";
14 end;
15 run;

We are using the %do_over() macro twice, once to create a list of data sets, second to construct the
select(); ... end; statement. With the help of options mprint; we can see that in the log:

the log - using "phrase=" parameter, ex. 1
1 1 %array(ds=sashelp.class, vars=age|, macarray=Y)
2 NOTE:[ARRAY] 6 macrovariables created
3 2
4 3 options mprint;
5 4 data
6 5 %do_over(age, phrase=%nrstr(data_%age(&_i_.)))
7 MPRINT(DO_OVER): data_15
8 MPRINT(DO_OVER): data_12
9 MPRINT(DO_OVER): data_16

10 MPRINT(DO_OVER): data_13
11 MPRINT(DO_OVER): data_14
12 MPRINT(DO_OVER): data_11
13 6 ;
14 7 set sashelp.class;
15 8
16 9 select(age);
17 10 %do_over(age, phrase=%nrstr(
18 11 when (%age(&_i_.)) output data_%age(&_i_.);
19 12 ))
20 MPRINT(DO_OVER): when (15) output data_15;
21 MPRINT(DO_OVER): when (12) output data_12;
22 MPRINT(DO_OVER): when (16) output data_16;
23 MPRINT(DO_OVER): when (13) output data_13;
24 MPRINT(DO_OVER): when (14) output data_14;
25 MPRINT(DO_OVER): when (11) output data_11;
26 13 otherwise put "unknown Age value!";
27 14 end;
28 15 run;
29 NOTE: There were 19 observations read from the data set SASHELP.CLASS.

25



B. Jabłoński - Macro Variable Arrays Made Easy with macroArray SAS Package

30 NOTE: The data set WORK.DATA_15 has 4 observations and 5 variables.
31 NOTE: The data set WORK.DATA_12 has 5 observations and 5 variables.
32 NOTE: The data set WORK.DATA_16 has 1 observations and 5 variables.
33 NOTE: The data set WORK.DATA_13 has 3 observations and 5 variables.
34 NOTE: The data set WORK.DATA_14 has 4 observations and 5 variables.
35 NOTE: The data set WORK.DATA_11 has 2 observations and 5 variables.

In the second example we want to run a user defined macro over several consecutive dates as that macro’s
parameter value:

code: using "phrase=" parameter, ex. 2
1 %array(date[4], function=202300 + _i_, macarray=Y)
2

3 %put %do_over(date);
4

5 %macro someMacro4(period);
6 data _null_;
7 put "Running macro for &period.";
8 run;
9 %mend someMacro4;

10

11 %do_over(date, phrase=%nrstr(
12 %someMacro4(%date(&_i_.))
13 ))

With help of the function= parameter in the %array() macro we are creating a list of dates (in the
"YYYYMM" form) which will be passed as arguments in %do_over() in order to use them as parameters in
the call to %someMacro4() macro:

the log - using "phrase=" parameter, ex. 2
1 1 %array(date[4], function=202300 + _i_, macarray=Y)
2 NOTE:[ARRAY] 4 macrovariables created
3 2
4 3 %put %do_over(date);
5 202301 202302 202303 202304
6 4
7 5 %macro someMacro4(period);
8 6 data _null_;
9 7 put "Running macro for &period.";

10 8 run;
11 9 %mend someMacro4;
12 10
13 11 %do_over(date, phrase=%nrstr(
14 12 %someMacro4(%date(&_i_.))
15 13 ))
16

17 Running macro for 202301
18 NOTE: DATA statement used (Total process time):
19 real time 0.00 seconds
20 cpu time 0.00 seconds
21

22 Running macro for 202302
23 NOTE: DATA statement used (Total process time):
24 real time 0.00 seconds
25 cpu time 0.00 seconds
26

27 Running macro for 202303
26

28 NOTE: DATA statement used (Total process time):
29 real time 0.01 seconds
30 cpu time 0.00 seconds
31

32 Running macro for 202304
33 NOTE: DATA statement used (Total process time):
34 real time 0.00 seconds
35 cpu time 0.00 seconds

Third example shows that inside the %do_over()’s phrases we can use multiple macro arrays at once:

code: using "phrase=" parameter, ex. 3
1 %array(ds=sashelp.class(obs=6), vars=name age, macarray=Y)
2

3 %do_over(name, phrase=%nrstr(
4 %put &_i_.. %name(&_i_.) has %age(&_i_.) years.;
5 ))

If two or more macro arrays have the same number of elements and "aligned" indexes, then they can be
used in the phrase= parameter text. One of those macros should be used as the first parameter of the
%do_over() to be a "driving" macro array (like the name MVA in the example).

the log - using "phrase=" parameter, ex. 3
1 1 %array(ds=sashelp.class(obs=6), vars=name age, macarray=Y)
2 NOTE:[ARRAY] 12 macrovariables created
3 2
4 3 %do_over(name, phrase=%nrstr(
5 4 %put &_i_.. %name(&_i_.) has %age(&_i_.) years.;
6 5 ))
7 1. Alfred has 14 years.
8 2. Alice has 13 years.
9 3. Barbara has 13 years.

10 4. Carol has 14 years.
11 5. Henry has 14 years.
12 6. James has 12 years.

The last parameter of the %do_over() is which=, this one expects a space separated list of "indexes
lists" over which the %do_over() will iterate. The idea behind this parameter is to allow subselection of
a macro array elements. A "indexes list" is a construction of the form start<:end<:by>>, examples of
such "indexes lists" are: 1:10:1, 1:10, 1:10:2, 1, 10, or even 10:1:-1. The which= allows multiple
"indexes lists" but sometimes single one is "better" i.e., it is better if multiple index lists can be condensed to
a single index list. For example, single "index list" of the form: 1:5 is equivalent to five one element "index
lists": 1 2 3 4 5, but use of the first one is more efficient under the hood. There are some technical
requirements about "indexes list": 1) default value of by is 1, 2) a single "index list" cannot have spaces
in its definition, 3) there a two special symbols: H and L which refers to higher and lower index values.
Interesting example where H, L, and -1 are used is to loop over a macro array backward:

code: using "which=" parameter
1 %array(test[*] x01-x06, vnames=Y, macarray=Y)
2

3 %put 1) %do_over(test);
4 %put 2) %do_over(test, which=H:L:-1);

The first loop from one to six, the second from six to one:

the log - using "which=" parameter
1 1 %array(test[*] x01-x06, vnames=Y, macarray=Y)
2 NOTE:[ARRAY] 6 macrovariables created
3 2
4 3 %put 1) %do_over(test);
5 1) x01 x02 x03 x04 x05 x06
6 4 %put 2) %do_over(test, which=H:L:-1);
7 2) x06 x05 x04 x03 x02 x01

Log printout from yet another example:

the log - using "which=" parameter, ex. 2
1 1 %array(test[*] x01-x99, vnames=Y, macarray=Y)
2 NOTE:[ARRAY] 99 macrovariables created
3 2
4 3 %put %do_over(test, which=L:3 97:H);
5 x01 x02 x03 x97 x98 x99

Macros %do_overN() and %make_do_over()
The %do_over2() and %do_over3() macros are utility macros which "extend" functionality of the
%do_over().

The %do_over() macro allows to loop over a macro array. The %do_over2() macro allows to loop over
two arrays by "loop-in-loop" approach, which gives us a Cartesian product of two arrays. The%do_over3()
macro allows to loop over three arrays by "loop-in-loop-in-loop" approach, which gives us a Cartesian product
of three arrays. Two following examples present use cases.

The first example shows how to run a macro with all possible combinations of input parameters values.

code: %do_over2() calling a two parameters macro
1 options nofullstimer nostimer; /* for shorter log */
2 %array(alpha[3] $ ("A" "B" "C"), macarray=Y)
3 %array( beta[3] ( 1 2 3 ), macarray=Y)
4

5 %macro letsPlay(x,y);
6 data &x&y;
7 x="&x.";
8 y= &y. ;
9 put x= y=;

10 run;
11 %mend letsPlay;
12

13 %do_over2(alpha, beta
14 ,phrase = %NRSTR(
15 %letsPlay(%alpha(&_I_.), %beta(&_J_))
16 ))

The log shows nine data sets created:

the log - nine data sets from %do_over2()
1 1 options nofullstimer nostimer; /* for shorter log */
2 2 %array(alpha[3] $ ("A" "B" "C"), macarray=Y)
3 NOTE:[ARRAY] 3 macrovariables created
4 3 %array( beta[3] ( 1 2 3 ), macarray=Y)
5 NOTE:[ARRAY] 3 macrovariables created
6 4
7 5 %macro letsPlay(x,y);
8 6 data &x&y;
9 7 x="&x.";

10 8 y= &y. ;
11 9 put x= y=;
12 10 run;
13 11 %mend letsPlay;
14 12
15 13 %do_over2(alpha, beta
16 14 ,phrase = %NRSTR(
17 15 %letsPlay(%alpha(&_I_.), %beta(&_J_))
18 16 ))
19

20 x=A y=1
21 NOTE: The data set WORK.A1 has 1 observations and 2 variables.
22

23 x=A y=2
24 NOTE: The data set WORK.A2 has 1 observations and 2 variables.
25

26 x=A y=3
27 NOTE: The data set WORK.A3 has 1 observations and 2 variables.
28

29 x=B y=1
30 NOTE: The data set WORK.B1 has 1 observations and 2 variables.
31

32 x=B y=2
33 NOTE: The data set WORK.B2 has 1 observations and 2 variables.
34

35 x=B y=3
36 NOTE: The data set WORK.B3 has 1 observations and 2 variables.
37

38 x=C y=1
39 NOTE: The data set WORK.C1 has 1 observations and 2 variables.
40

41 x=C y=2
42 NOTE: The data set WORK.C2 has 1 observations and 2 variables.
43

44 x=C y=3
45 NOTE: The data set WORK.C3 has 1 observations and 2 variables.

The second example shows Cartesian product of a single macro array with itself and itself... so three times.

code: %do_over3() producing Cartesian product over macro array A
1 %array(a[2] (0 1), macarray=Y)
2

3 %do_over3(a, a, a
4 , phrase = %NRSTR(
5 %put sum(%a(&_I_.),%a(&_J_),%a(&_K_))=%sysevalf(%a(&_I_.)+%a(&_J_)+%a(&_K_));
6 ))

In the log we can see:

the log - all possible sums
1 1 %array(a[2] (0 1), macarray=Y)
2 NOTE:[ARRAY] 2 macrovariables created
3 2
4 3 %do_over3(a, a, a
5 4 , phrase = %NRSTR(
6 5 %put sum(%a(&_I_.),%a(&_J_),%a(&_K_))=%sysevalf(%a(&_I_.)+%a(&_J_)+%a(&_K_));
7 6 ))
8 sum(0,0,0)=0
9 sum(0,0,1)=1

10 sum(0,1,0)=1
11 sum(0,1,1)=2
12 sum(1,0,0)=1
13 sum(1,0,1)=2
14 sum(1,1,0)=2
15 sum(1,1,1)=3

As we can notice both macros have dedicated internal looping variables. For the %do_over2() macro they
are _I_ and _J_, and for The %do_over3() macro we also have _K_. The %do_over*() macros have
no which= parameter for them, only phrase= and between= are available.

With fairly high probability we can say that 99% of practical situations of looping over macro arrays can
be done with %do_over()(80%), %do_over2()(13%), and with %do_over3()(6%), but there is still
this 1% of "special situations" where a "do over of higher order" may be required (i.e. greater than 3 nested
loops). That is the reason why the %make_do_over() macro was implemented. Since it is rather "special
case" situation we only look at logs from two small examples.

the log - size must be greater than 3
1 1 %make_do_over(2);
2 NOTE:[MAKE_DO_OVER] NO MACRO GENERATED FOR SIZE = 2
3 [MAKE_DO_OVER] SIZE must be greater than 3!!!

the log - loop of size five
1 1 %make_do_over(5);
2

3 NOTE: The file _******_ is:
4 Filename=*******************************,
5 RECFM=V,LRECL=512,File Size (bytes)=0,
6 Last Modified=19May2024:12:34:56,
7 Create Time=19May2024:12:34:56
8 NOTE: 41 records were written to the file _******_.
9 The minimum record length was 1.

10 The maximum record length was 64.
11 NOTE: Fileref _******_ has been deassigned.
12 43
13 44 %array(a5_[2] (0 1), macarray=Y)
14 NOTE:[ARRAY] 2 macrovariables created
15 45
16 46 %do_over5(a5_, a5_, a5_, a5_, a5_
17 47 ,phrase = %NRSTR(
18 48 %put (%a5_(&_I1_.),%a5_(&_I2_),%a5_(&_I3_),%a5_(&_I4_),%a5_(&_I5_));
19 49 ))
20 (0,0,0,0,0)

21 (0,0,0,0,1)
22 ...
23 ...
24 ...
25 (1,1,1,1,0)
26 (1,1,1,1,1)

What is worth mentioning, the loop index variables are in this case _I1_, _I2_, _I3_, etc. All technical
details are described in the documentation of the macro. As mentioned earlier, all the documentation is
available at macroArray package’s repository.

Macros %zipArrays() and %QzipArrays()
Two final macros from the macro arrays suite in the macroArray package are dedicated to execute opera-
tions and functions on respective elements of two macro arrays and produce a new macro array with results
of processing. The only difference between them is that the %QzipArrays()returns macro quoted results
and %zipArrays() does not. The "zip" in the macro name has nothing to do with data compression, its
rather a reference to a "zipper", where corresponding pairs of teeth from the left and right row are "meshed"
together. When one of macro arrays has fewer elements than the other then elements of the shorter are, by
default, "reused" starting from the beginning. This behavior can be altered, to produce Cartesian product or
stop at shorter, with help of the reuse= parameter.

Two positional parameters, which are required, are names of macro arrays. The default function operating
on corresponding elements is the cat() function, the name of newly created macro array is concocted
from names of input macro arrays. Both the function and the name can be altered, respectively, by the
function= and the result= parameter. Also the macarray= parameter (known from the %array()
macro) is available.

The following example creates new macro array ab with concatenated values of macro arrays a and b (values
of b are reused):

code: simple zip of two macro arrays
1 %array(a[4] X Y Z T, vnames=Y)
2 %array(b[2] (1 2))
3

4 %put _user_;
5 %zipArrays(a, b);
6 %put _user_;

The log reveals the following information:

the log - new macro array AB highlighted
1 1 %array(a[4] X Y Z T, vnames=Y)
2 NOTE:[ARRAY] 4 macrovariables created
3 2 %array(b[2] (1 2))
4 NOTE:[ARRAY] 2 macrovariables created
5 3
6 4 %put _user_;
7 GLOBAL A1 X
8 GLOBAL A2 Y
9 GLOBAL A3 Z

10 GLOBAL A4 T
11 GLOBAL AHBOUND 4
12 GLOBAL ALBOUND 1

13 GLOBAL AN 4
14 GLOBAL B1 1
15 GLOBAL B2 2
16 GLOBAL BHBOUND 2
17 GLOBAL BLBOUND 1
18 GLOBAL BN 2
19 5 %zipArrays(a, b);
20 NOTE:[ZIPARRAYS] 4 macrovariables created
21 6 %put _user_;
22 GLOBAL A1 X
23 GLOBAL A2 Y
24 GLOBAL A3 Z
25 GLOBAL A4 T
26 GLOBAL AB1 X1
27 GLOBAL AB2 Y2
28 GLOBAL AB3 Z1
29 GLOBAL AB4 T2
30 GLOBAL ABHBOUND 4
31 GLOBAL ABLBOUND 1
32 GLOBAL ABN 4
33 GLOBAL AHBOUND 4
34 GLOBAL ALBOUND 1
35 GLOBAL AN 4
36 GLOBAL B1 1
37 GLOBAL B2 2
38 GLOBAL BHBOUND 2
39 GLOBAL BLBOUND 1
40 GLOBAL BN 2

This example creates new macro array and utilizes parameters presented above (new macro array is Cartesian
product in this case):

code: zip of two macro arrays with parameters
1 %array(a[2] (100 200))
2 %array(b[2] (3 4))
3

4 %put _user_;
5 %zipArrays(a, b
6 , result=NEW_ARR
7 , function=SUM
8 , reuse=CP
9 , macarray=Y

10 )
11 %put _user_;
12

13 %put %do_over(NEW_ARR);

The log shows everything, even additional note for the macarray= parameter:

the log - zip of two macro arrays with parameters
1 1 %array(a[2] (100 200))
2 NOTE:[ARRAY] 2 macrovariables created
3 2 %array(b[2] (3 4))
4 NOTE:[ARRAY] 2 macrovariables created
5 3
6 4 %put _user_;
7 GLOBAL A1 100

8 GLOBAL A2 200
9 GLOBAL AHBOUND 2

10 GLOBAL ALBOUND 1
11 GLOBAL AN 2
12 GLOBAL B1 3
13 GLOBAL B2 4
14 GLOBAL BHBOUND 2
15 GLOBAL BLBOUND 1
16 GLOBAL BN 2
17 5 %zipArrays(a, b
18 6 , result=NEW_ARR
19 7 , function=SUM
20 8 , reuse=CP
21 9 , macarray=Y
22 10 )
23 NOTE:[ZIPARRAYS] 4 macrovariables created
24

25 NOTE: When macarray= parameter is active the zipArrays macro
26 cannot be called within the %put statement.
27 NOTE: Execution like: %put %zipArrays(..., macarray=Y) will
28 result with an e.r.r.o.r.
29 11 %put _user_;
30 GLOBAL A1 100
31 GLOBAL A2 200
32 ...
33 GLOBAL BLBOUND 1
34 GLOBAL BN 2
35 GLOBAL NEW_ARR1 103
36 GLOBAL NEW_ARR2 104
37 GLOBAL NEW_ARR3 203
38 GLOBAL NEW_ARR4 204
39 GLOBAL NEW_ARRHBOUND 4
40 GLOBAL NEW_ARRLBOUND 1
41 GLOBAL NEW_ARRN 4
42 12
43 13 %put %do_over(NEW_ARR);
44 103 104 203 204

See the documentation where all remaining details about additional parameters, not mentioned here, are
given. Also many more examples are presented there.

The %mcDictionary()macro
Regarding the list of questions that were presented at the beginning of this paper that led to the development
of the macroArray package, one more question can be added: Does the "suffix" (array’s index) have to
always be an integer? Can we, instead writing for example%myArray(1), write%myDictionary(Bart)
and get the value, let’s say "180cm", extracted for key "Bart"? Yes, the macroArray package also pro-
vides such functionality. Let’s talk about macro dictionaries.

Macro dictionary (aka MD) is a concept similar to the macro variable array, but instead "integer-value" pairs
like MVAs have, it allows for a "key-value" relations between macro variables values. The "key" in this case can
be arbitrary text string associated with a value and the value is retrieved from macro dictionary by providing
the key. In the essence, the MD behavior is similar to the way formats or hash tables behaves (though the
mechanism under the hood is different). Worth to mention is that a macro dictionary works according to the
"one key - one value" assumption.

The macroArray package facilitates macro dictionaries by the %mcDictionary() macro. A dictionary
is created by calling the %mcDictionary() macro with a name to be used by the dictionary. The naming
convention is "SAS standard" with the respect to the following restrictions: cannot be empty, cannot be
underscore, cannot be longer than 13 symbols. First, we create an empty macro dictionary named myDict:

code: simple (and empty) dictionary
1 %mcDictionary(myDict)

The log does not show us anything interesting but it does not mean nothing happened. In fact quite a lot
has happened. Under the hood a macro named %myDict() was generated, similar way when we used
macarray= parameter in the %array() macro, which instantiate the MD and its use. The %myDict()
macro, and also any other macro created by the %mcDictionary()macro, has three parameters. One po-
sitional and two "key-value" type. The first, positional parameter named method, is to tell the %myDict()
macro what work to execute in particular under the hood. It accepts the following list of values: LIST [L],
ADD [A], FIND [F], CHECK [C], DEL [D], and CLEAR (in square brackets are aliases). We discuss them in a
moment, but let’s finish describing parameters first. The "key-value" parameters are: key= and data= and
whether they are used or not depends on the value of the method parameter.

Let’s talk about the method parameter values (also referred simply as "methods"). When method is empty
then a table with list of accepted values is printed in the log.

The first method we discuss is the LIST [L], which makes %myDict() to print out in the log all its elements
(keys, values, and hash digest associated with key). In our case since the %myDict() has no elements yet,
the printout will be just:

the log - list of empty dictionary
1 1 %myDict(List)
2

3 Content of the dictionary:
4

5 --------------------------

To populate our MD with new data we use the ADD [A] method (this is one of two ways of populating MD with
data). When we use ADD [A] the key= provides value for the key in the MD and the data= provides value
for the data portion in the MD. For example, the following code adds three entries to the %myDict() MD we
created:

code: adding data to MD
1 %myDict(ADD,key=A,data=I)
2 %myDict(A ,key=B,data=<3)
3 %myDict(A ,key=C,data=SAS)
4 %myDict(L)

And in the log we can see:

the log - three entries and listing
1 1 %myDict(ADD,key=A,data=I)
2 2 %myDict(A ,key=B,data=<3)
3 3 %myDict(A ,key=C,data=SAS)
4 4 %myDict(L)
5

6 Content of the dictionary:
7

8 key=C
9 hash=0D61F8370CAD1D41

10 val=SAS
11

12 key=A
13 hash=7FC56270E7A70FA8
14 val=I
15

16 key=B
17 hash=9D5ED678FE57BCCA
18 val=<3
19

20 --------------------------

As we can see, the process of adding data to the MD does not reveal too much but yields of the process can
be verified by the LIST [L] method.

When the MD is not empty we can try to use the data. This can be done in two ways. The first, we can just
check if a given key exists in the MD, the CHECK [C] method does that. For example, the following code
returns zeros and ones of existing and non existing keys:

code: checking keys
1 %put
2 A:%myDict(CHECK,key=A)
3 B:%myDict(C,key=B)
4 C:%myDict(C,key=C)
5 D:%myDict(C,key=D)
6 ;

The log shows:

the log - checking keys
1 A:1 B:1 C:1 D:0

The second, we can for a given key extract associated data portion, the FIND [F] method does that. For
example, to get: "I ♡ SAS" in the log we run:

code: collecting data values
1 %put
2 "%myDict(FIND,key=A)
3 %myDict(F,key=B)
4 %myDict(F,key=C)
5 %myDict(F,key=D)"
6 ;

The log shows (mind there is a space character at the beginning of lines 3, 4, and 5 in the code):

the log - collecting data values
1 "I <3 SAS "

As we can see the value of not existing key "D" is empty (the space is the one from between the bracket and
the percent in ...=C) %myD...).

35



B. Jabłoński - Macro Variable Arrays Made Easy with macroArray SAS Package

We discuss two final methods in one thread since both are dedicated to removing data from a MD. The methods
are: DEL [D], and CLEAR. The difference between them is that the DEL [D] removes data for only a particular
key (so key= parameter is needed) and the CLEAR just deletes them all and CLEAR does not have an alias.
Let’s see it on the following example:

code: deleting entries from MD
1 %myDict(DEL,key=B)
2 %myDict(L)
3

4 %myDict(CLEAR)
5 %myDict(L)

In the log we can see:

the log - deleting entries from MD
1 1 %myDict(DEL,key=B)
2 2 %myDict(L)
3

4 Content of the dictionary:
5

6 key=C
7 hash=0D61F8370CAD1D41
8 val=SAS
9

10 key=A
11 hash=7FC56270E7A70FA8
12 val=I
13

14 --------------------------
15 3
16 4 %myDict(CLEAR)
17 5 %myDict(L)
18

19 Content of the dictionary:
20

21 --------------------------

In both cases, under the hood macro variables of the form myDict_****************_K and
myDict_****************_V are looked up, the "****************" indicates a sixteen sym-
bols of alphanumeric hash digest of the key. But for the DEL [D] method only "that one" is deleted, while the
CLEAR methods removes them all. As a side note, by looking at the internal representation of MD variables,
we now know why the naming convention limits the name to 13 characters (32 - (16+3)).

When we were discussing the ADD method a second method of populating MD with data was mentioned. The
method, similarly as in the macro arrays case is based on taking data from the data set. To use data set we
have to indicate the fact when we execute the %mcDictionary() macro. Three dedicated parameters:
ds=, k=, and d= can be use for the process. The ds= indicates data set, the k= indicates key variable (one
variable!) in the data set, and the d= indicates data portion variable (also one variable!) in the data set.
Default values for k= and d= are "Key" and "Data" so if the data set contains such variables and we want use
them we can skip k= and d=. The following code present an example:

code: populating MD from data set
1 data work.have;
2 input kVar :$1. dVar :$3.;
3 cards;

4 A I
5 B <3
6 C SAS
7 ;
8 run;
9

10 %mcDictionary(myDSdict, DCL, ds=work.have, k=kVar, d=dVar)
11

12 %myDSdict(L)

Notes in the log confirm everything:

the log - populating MD from data set
1 1 data work.have;
2 2 input kVar :$1. dVar :$3.;
3 3 cards;
4 NOTE: The data set WORK.HAVE has 3 observations and 2 variables.
5 ...
6 7 ;
7 8 run;
8 9
9 10 %mcDictionary(myDSdict, DCL, ds=work.have, k=kVar, d=dVar)

10 NOTE:[MCDICTIONARY] Populating dictionary myDSdict.
11 NOTE:[MCDICTIONARY] -----------------------------
12 11
13 12 %myDSdict(L)
14

15 Content of the dictionary:
16

17 key=C
18 hash=0D61F8370CAD1D41
19 val=SAS
20

21 key=A
22 hash=7FC56270E7A70FA8
23 val=I
24

25 key=B
26 hash=9D5ED678FE57BCCA
27 val=<3
28

29 --------------------------

Except the parameters mentioned above, one more thing can be observed in call to%mcDictionary(...,
DCL,...) macro. The DCL value (also DECLARE or just null) in the second parameter tells the
%mcDictionary() macro that either a macro dictionary is to be created (when parameter is DCL) or
a macro dictionary is to be deleted (when parameter is DELETE). If the DELETE is used not only MD’s macro
variables are deleted but the MD macro itself is removed from the SAS session.

EXAMPLE

In the last section we will take a look at some more "advanced" examples that illustrate the use of both macro
arrays and macro dictionaries.

As the first step we will prepare "driving" data sets:
code: creating "driving" data sets

1 /* project driving data */
2 /* data set with list of functions to "run over data" */
3 data work.functions;
4 input fName $12.;
5 cards;
6 sum
7 mean
8 median
9 min

10 max
11 nmiss
12 std
13 range
14 stderr
15 var
16 ;
17 run;
18

19 /* data set with project metadata */
20 data work.projectMetadata;
21 infile cards dsd dlm=",";
22 input key :$16. data :$128.;
23 cards;
24 ID,ABC-123-XYZ
25 TITLE,Use case of the MacroArray package
26 PATH,/path/to/study/data
27 INDATASET,sashelp.cars
28 OUTDATASET,work.results
29 VARIABLE,invoice
30 GROUPBY,origin
31 STARTDT,2020-01-01
32 ENDDT,2024-12-31
33 ;
34 run;

The log confirms data sets creation:
the log - creating "driving" data sets

1 1 /* project driving data */
2 2 /* data set with list of functions to "run over data" */
3 3 data work.functions;
4 4 input fName $12.;
5 5 cards;
6

7 NOTE: The data set WORK.FUNCTIONS has 10 observations and 1 variables.
8 ...
9

10 16 ;
11 17 run;
12 18
13 19 /* data set with project metadata */
14 20 data work.projectMetadata;
15 21 infile cards dsd dlm=",";

16 22 input key :$16. data :$128.;
17 23 cards;
18

19 NOTE: The data set WORK.PROJECTMETADATA has 9 observations and 2 variables.
20 ...
21

22 33 ;
23 34 run;

The driving data allows us to execute the "project code" and the use of macro array and macro dictionary
makes the project code only "data driven". Here is the code:

code: data driven project
1 /* project code */
2 /* create macro array FN and macro dictionary PRJ */
3 %array(ds=work.functions, vars=fName#fN, macarray=Y)
4 %mcDictionary(prj, DCL, DS=work.projectMetadata)
5

6 %put _user_;
7

8 title1 "Title: %prj(F,key=TITLE)";
9 title2 "Project %prj(F,key=ID), located at: %prj(F,key=PATH)";

10 title3 "starts %prj(F,key=STARTDT) and ends %prj(F,key=ENDDT)";
11 footnote1 "Input data set: %prj(F,key=INDATASET)";
12 footnote2 "Output data set: %prj(F,key=OUTDATASET)";
13 footnote3 "Analyzed variable: %prj(F,key=VARIABLE)";
14 /* check if the grouping variable exists */
15 footnote4
16 %if %prj(C,key=GROUPBY) %then
17 %do; "Analysis in groups by: %prj(F,key=GROUPBY)" %end;
18 ;
19

20 /* aggregate data */
21 Proc SQL;
22 create table %prj(F,key=OUTDATASET) as
23 select
24 /* check if the grouping variable exists */
25 %if %prj(C,key=GROUPBY) %then %do; %prj(F,key=GROUPBY), %end;
26 /* loop over aggregating functions */
27 %do_over(fN
28 ,phrase=%NRSTR(
29 /* apply function to analysisVariable
30 and name the result "analysisVariable_functionName" */
31 %fN(&_i_.)(%prj(F,key=VARIABLE)) as %prj(F,key=VARIABLE)_%fN(&_i_.)
32 )
33 ,between=%str(,)
34 )
35

36 from
37 %prj(F,key=INDATASET)
38

39 /* check if the grouping variable exists */
40 %if %prj(C,key=GROUPBY) %then
41 %do;
42 group by
43 %prj(F,key=GROUPBY)
44 %end;
45 ;
46 Quit;

47

48 /* print data */
49 proc print data = %prj(F,key=OUTDATASET) ;
50 run;
51

52 title;
53 footnote;
54

55 /* end of project code */

The log from the process is the following:
the log - data driven project

1 1 /* project code */
2 2 /* create macro array FN and macro dictionary PRJ */
3 3 %array(ds=work.functions, vars=fName#fN, macarray=Y)
4 NOTE:[ARRAY] 10 macrovariables created
5 4 %mcDictionary(prj, DCL, DS=work.projectMetadata)
6 NOTE:[MCDICTIONARY] Populating dictionary prj.
7 NOTE:[MCDICTIONARY] -----------------------------
8 5
9 6 %put _user_;

10 GLOBAL FN1 sum
11 GLOBAL FN2 mean
12 GLOBAL FN3 median
13 GLOBAL FN4 min
14 GLOBAL FN5 max
15 GLOBAL FN6 nmiss
16 GLOBAL FN7 std
17 GLOBAL FN8 range
18 GLOBAL FN9 stderr
19 GLOBAL FN10 var
20 GLOBAL FNHBOUND 10
21 GLOBAL FNLBOUND 1
22 GLOBAL FNN 10
23 GLOBAL PRJ_5FFB5F0D0DE78321_K PATH
24 GLOBAL PRJ_5FFB5F0D0DE78321_V /path/to/study/data
25 GLOBAL PRJ_6F9DCCD85B2E0786_K TITLE
26 GLOBAL PRJ_6F9DCCD85B2E0786_V Use case of the MacroArray package
27 GLOBAL PRJ_815250411C5E7F5E_K STARTDT
28 GLOBAL PRJ_815250411C5E7F5E_V 2020-01-01
29 GLOBAL PRJ_97C6507873878CEA_K GROUPBY
30 GLOBAL PRJ_97C6507873878CEA_V origin
31 GLOBAL PRJ_9B54A6B75069B38A_K OUTDATASET
32 GLOBAL PRJ_9B54A6B75069B38A_V work.results
33 GLOBAL PRJ_AC8C59A0734680FB_K ENDDT
34 GLOBAL PRJ_AC8C59A0734680FB_V 2024-12-31
35 GLOBAL PRJ_B718ADEC73E04CE3_K ID
36 GLOBAL PRJ_B718ADEC73E04CE3_V ABC-123-XYZ
37 GLOBAL PRJ_D519071E56F75CF8_K INDATASET
38 GLOBAL PRJ_D519071E56F75CF8_V sashelp.cars
39 GLOBAL PRJ_E61AD9B2553A293B_K VARIABLE
40 GLOBAL PRJ_E61AD9B2553A293B_V invoice
41 GLOBAL PRJ_KEYSNUM 0
42

43 7
44 8 title1 "Title: %prj(F,key=TITLE)";
45 9 title2 "Project %prj(F,key=ID), located at: %prj(F,key=PATH)";
46 10 title3 "starts %prj(F,key=STARTDT) and ends %prj(F,key=ENDDT)";
47 11 footnote1 "Input data set: %prj(F,key=INDATASET)";

48 12 footnote2 "Output data set: %prj(F,key=OUTDATASET)";
49 13 footnote3 "Analyzed variable: %prj(F,key=VARIABLE)";
50 14 /* check if the grouping variable exists */
51 15 footnote4
52 16 %if %prj(C,key=GROUPBY) %then
53 17 %do; "Analysis in groups by %prj(F,key=GROUPBY)" %end;
54 18 ;
55 19
56 20 /* aggregate data */
57 21 Proc SQL;
58 22 create table %prj(F,key=OUTDATASET) as
59 MPRINT(PRJ): work.results
60 23 select
61 24 /* check if the grouping variable exists */
62 25 %if %prj(C,key=GROUPBY) %then %do; %prj(F,key=GROUPBY), %end;
63 MPRINT(PRJ): origin
64 26 /* loop over aggregating functions */
65 27 %do_over(fN
66 28 ,phrase=%NRSTR(
67 29 /* apply function to analysisVariable
68 30 and name the result "analysisVariable_functionName" */
69 31 %fN(&_i_.)(%prj(F,key=VARIABLE)) as %prj(F,key=VARIABLE)_%fN(&_i_.)
70 32 )
71 33 ,between=%str(,)
72 34 )
73 MPRINT(DO_OVER): sum(invoice) as invoice_sum,
74 MPRINT(DO_OVER): mean(invoice) as invoice_mean,
75 MPRINT(DO_OVER): median(invoice) as invoice_median,
76 MPRINT(DO_OVER): min(invoice) as invoice_min,
77 MPRINT(DO_OVER): max(invoice) as invoice_max,
78 MPRINT(DO_OVER): nmiss(invoice) as invoice_nmiss,
79 MPRINT(DO_OVER): std(invoice) as invoice_std,
80 MPRINT(DO_OVER): range(invoice) as invoice_range,
81 MPRINT(DO_OVER): stderr(invoice) as invoice_stderr,
82 MPRINT(DO_OVER): var(invoice) as invoice_var
83 35
84 36 from
85 37 %prj(F,key=INDATASET)
86 MPRINT(PRJ): sashelp.cars
87 38
88 39 /* check if the grouping variable exists */
89 40 %if %prj(C,key=GROUPBY) %then
90 41 %do;
91 42 group by
92 43 %prj(F,key=GROUPBY)
93 MPRINT(PRJ): origin
94 44 %end;
95 45 ;
96 NOTE: Table WORK.RESULTS created, with 3 rows and 11 columns.
97

98 46 Quit;
99 NOTE: PROCEDURE SQL used (Total process time):

100 real time 0.03 seconds
101 cpu time 0.03 seconds
102

103 47
104 48 /* print data */
105 49 proc print data = %prj(F,key=OUTDATASET) ;
106 MPRINT(PRJ): work.results

107 50 run;
108

109 NOTE: There were 3 observations read from the data set WORK.RESULTS.
110 NOTE: PROCEDURE PRINT used (Total process time):
111 real time 0.01 seconds
112 cpu time 0.01 seconds
113

114 51
115 52 title;
116 53 footnote;
117 54
118 55 /* end of project code */

The code, with help of macro array and macro dictionary, generated from data sets is fully data driven. No
changes in the code are needed, user need only modify driving data sets to either change list of aggregating
functions or the project itself. And the output looks like this:

CONCLUSION

In the article, we learned how to simplify the problem of creating SAS macro variable arrays with a help of
the macroArray package. But before "the end" two more things require emphasis!

The first, for advanced SAS programmer, experienced in the art of "multi-ampersanding", who fully under-
stand the process of indirect referencing, you can still benefit from the package by making your code eas-
ier to read and more straightforward in some cases. You can enjoy "syntactic sugar" of the %array()
and %mcDictionary() macros the same way you enjoy the IF-subsetting (if <condition>;) or the
SUM-statement (variable + <expression>;).

The second, for a "junior" SAS programmer. The package is design to support inexperienced users walking
their "multi-ampersands" path and the author encourages them to use the macroArray package for their
benefit, but... at the same time the author highly recommends you to put an effort and attention in a deep
understanding of "how does it work?". End then, when understood, to check the previous paragraph ;-)

The End

Appendix Appendix A - code coloring guide

The best experience for reading this article is in color and the following convention is used:

 The code snippets use the following coloring convention:

code: is surrounded by a black frame
1 In general we use black ink for the code but:
2 - code of interest is in red ink so that it can be highlighted,
3 - and comments pertaining to code are in a bluish ink for easier reading.


 The LOG uses the following coloring convention:
the log - is surrounded by a blueish frame

1 The source code and general log text are blueish.
2 Log NOTEs are green.
3 Log WARNINGSs are violet.
4 Log ERRORSs are red.
5 Log text generated by the user is purple.

Appendix Appendix B - install the SPF and the package

To install the SAS Packages Framework and a SAS Package we execute the following steps:

 First we create a directory to install SPF and Packages, for example: /home/user/packages or
C:/packages.


 Next, depending if the SAS session has access to the internet:
○ if it does - we run the following code:

code: install from the internet
1 filename packages "/home/user/packages";
2

3 filename SPFinit url
4 "https://raw.githubusercontent.com/yabwon/SAS_PACKAGES/main/SPF/SPFinit.sas";
5 %include SPFinit;
6

7 %installPackage(SPFinit)
8 %installPackage(macroArray)

○ If the SAS session does not have access to the internet we go to the framework repository:
https://github.com/yabwon/SAS_PACKAGES

next (if not already) we click the stargazer button [☀] ;-) and then we navigate to the SPF directory and
we copy the SPFinit.sas file into the directory from step one (direct link:
https://raw.githubusercontent.com/yabwon/SAS_PACKAGES/main/SPF/SPFinit.sas
). And for packages - we just copy the package zip file into the directory from step one.


 From now on, in all subsequent SAS session, it is enough to just run:
code: enable framework and load packages

1 filename packages "/home/user/packages";
2 %include packages(SPFinit.sas);
3 %loadPackage(macroArray)

to enable the framework and load packages. To update the framework or a package to the latest version
we simply run:

code: update from the internet
1 %installPackage(SPFinit macroArray)

Appendix Appendix C - safety considerations

The SPF installation process, in a "nutshell", reduces to copying the SPFinit.sas file into the packages
directory. It is the same for a packages too.

You may ask: is it safe to install?

Yes, it’s safe! When you install the SAS Packages Framework, and later when you install packages, the files
are simply copied into the packages directory that you configured above. There are no changes made to
your SAS configuration files, or autoexec, or registry, or anything else that could somehow "break SAS." As
you saw, you can perform a manual installation simply by copying the files yourself. Furthermore the SAS
Packages Framework is:

 written in 100% SAS code, it does not require any additional external software to work,

 full open source (and MIT licensed), so every macro can be inspected.

When we work with a package, before we even start thinking about loading content of one into the SAS
session, both the help information and the source code preview are available.

To read help information (printed in the log) you simply run:
code: get help info

1 %helpPackage(<packageName>, <*|componentName|license>)

To preview source code of package components (also printed in the log) you simply run:
code: get code preview

1 %previewPackage(<packageName>, <*|componentName>)

The asterisk means "print everything", the componentName is the name of a macro, or a function, or a
format, etc. you want see.

INDEX

concept
INDIRECT REFERENCE, 2, 3
MACRO ARRAY, 2
MACRO DICTIONARY, 2, 33
MACRO VARIABLE ARRAY, 2
MVA, 2

function
DOSUBL(), 6, 8

